
Type: legacy re-engineering

Language: Fortran

Compiler: gfortran

Skills: program re-engineering

Experience Level: novice-intermediate

“The determined Real Programmer can write FORTRAN programs in any language.”
Ed Post.



INTRODUCTION

In 1966, Lawrence Fisher described a Fortran program to calculate exact rates of  return1. He 
describes the algorithm as “solving what is, perhaps, the most complex of  all common, compound-interest 
problems: the determination of  a realized rate of  return on an investment portfolio where funds have been added to 
and withdrawn from the portfolio after the initial investment”.

We take it for granted now, but in 1966 calculating such a numeric value was, as Fisher described 
it, a “tedious” task. Interestingly the algorithm is described as being so efficient that the “machine 
time” used in finding the rate of  return “costs less than one cent”. 

1 Fisher, L., “An algorithm for finding exact rates of return”, The Journal of Business, Vol.39(1), pp.
111-118 (1966)



THE CODE

The program itself  is coded in Fortran II.

      DIMENSION T, DIFR(999), P(999), CASE(9)
      LIMIT = 100
  10  READ INPUT TAPE 5, 20, MONTH, DAY, YEAR, V, N, CASE
  20  FORMAT (I2, 2(1XF2.0), F11.2, I3, 9A6)
      BIGT = TIME(MONTH, DAY, YEAR)
      DO 30 J = 1, N
      READ INPUT TAPE 5, 20, MONTH, DAY, YEAR, P(J)
  30  T DIFR(J)= BIG T - TIME(MONTH, DAY, YEAR)
      SMALL I = 0.
      DO 200 K = 1, LIMIT
      SUM VAL = V
      DENOM = 0.
      DO 100 J = 1, N
      PRSNTV = P(J) * EXPF(SMALL I * T DIFR(J))
      SUM VAL = SUM VAL - PRSNTV
 100  DENOM = DENOM + TDIFR(J) * PRSNTV
      DELTA I = SUM VAL /DENOM
      SMALL I = SMALL I + DELTA I
      IF (ABSF(DELTA I) - 0.00005) 300, 300, 200
 200  CONTINUE
      WRITE OUTPUT TAPE 6, 210, CASE
      GO TO 10  
 210  FORMAT (17H FAILURE ON CASE 9A6)
 300  R = EXPF(SMALL I) - 1.0
      WRITE OUTPUT TAPE 6, 320, SMALL I, R, CASE
      GO TO 10
 320  FORMAT (1H 2F8.5, 9A6//)
      END(1,0,0,0,0,0,1,0,0,1,0,0,0,0,0)     

It uses the library function EXPF to calculate ex, and an external function TIME for converting 
time floating-point numbers (written in FAP - Fortran Assembly program). 

See listing rates_1.for



DISASSEMBLE AND ANALYZE THE PROGRAM

The first task is to assess the program, by disassembling it. Assume no prior knowledge, the task 
here is to re-engineer the program, not create it from scratch. Don’t be fooled by the programs 
mere 28 lines of  code. Re-engineering a program can be a challenging experience, no matter the 
size of  the program.

Firstly, this program won’t run. Many programs written in Fortran 77 can be compiled because 
gfortran is backwards compatible... to a point. This program is written in Fortran II for the IBM 
7094, an enhanced derivative of  the Fortran compiler originally written for the IBM 704 by John 
Backus and his team.

A legacy program such as this one contains a number of  structures which are likely obsolete. The 
first thing to do with such a program is to analyze the code to determine what the core 
components are. 

• There is input from and output to a file in the form of  a tape spool.
• There is a loop associated with reading input data from the tape.
• There are two nested loops which performs the calculation.
• There is an arithmetic IF statement.
• There are two GO TO statements.
• There is use of  an external function, TIME (FAP-coded).
• The program is not free-format, and in uppercase.
• The program uses the descriptive terms BIG, SMALL, SUM, DELTA.
• The program uses Fortran II functions - EXPF, and ABSF.

The challenge in re-engineering even such a small program is where to start? The easiest place to 
start may be to deal with improving readability. Next decaying code such as the tape I/O will be 
dealt with. Next the assembler code for the external function TIME will be translated into 
Fortran and incorporated into the program. 

IMPROVE READABILITY

To improve readability the first task is to convert the program from uppercase to lowercase. This 
can be achieved using the “translate characters” utility, tr:

! tr '[:upper:]' '[:lower:]' < input.txt > output.txt

The terms BIG, SMALL etc. are removed

See listing rates_2.for



RE_ENGINEERING INPUT AND OUTPUT

The first thing to note about this program is that there are a series of  I/O statements which 
relate to reading from and writing to TAPE, along with associated FORMAT statements. 

  10  read input tape 5, 20, month, day, year, v, n, case
  20  format (I2, 2(1XF2.0), F11.2, I3, 9A6)
      read input tape 5, 20, month, day, year, p(j)
      write output tape 6, 210, case
 210  format (17H Failure on case 9A6)
      write output tape 6, 320, I, R, case
 320  format (1H 2F8.5, 9A6//)

At 7 lines, I/O comprises 25% of  the program. Even if  you could get the program to compile, it 
is unlikely you have a tape device attached to your machine. These statements will be flagged to 
be modified. In the case of  the read statements, the TAPE 5 refers to the ID of  the tape to be 
read from, and similarly TAPE 6 refers to tape drive 6, which is to be written to. The values 20 
and 210/320 refer to FORMAT statements, and the remaining parameters refer to the actual 
variables. The question of  how these statements should be modified depends on the type of  data 
that is being input. For trivial amounts of  data the best idea would be to convert the statements to 
console I/O, large amounts of  data should be converted to file I/O. This is better understood by 
looking at the data involved. A sample input is shown below:

! 10/31/65!   108.00  1  Example 1
!  1/12/65    100.00
     12/31/65    132.00  2  Example 2
     12/31/64   -100.00
! 12/31/63    200.00  
      1/ 1/66   1000.00 11  Example 3--A 4 PERCENT BOND
      1/ 1/66    -20.00
      7/ 1/61    -20.00
      1/ 1/62    -20.00
      7/ 1/62    -20.00
      1/ 1/63    -20.00
      7/ 1/63    -20.00
      1/ 1/64    -20.00
      7/ 1/64    -20.00
      1/ 1/65    -20.00
      7/ 1/65    -20.00
      1/ 1/61   1000.00

The input is likely best converted to file input. The output on the other hand, can be directed to 
standard output.

      0.09620 0.10098 Example 1
      0.09531 0.10000 Example 2
      0.03961 0.04040 Example 3--A 4 PERCENT BOND
    



The output suffers somewhat from usability issues, but those can be dealt with after the program 
has been re-engineered and is functioning.

The first READ statement:

  10  read input tape 5, 20, month, day, year, v, n, case

reads from “input tape 5”, using the format statement at label 20, values into six different 
variables. To put this in context, consider the first line of  the input:

! 10/31/65!   108.00  1  Example 1

After the input:

! month = 10
! day   = 31
! year  = 65 
! v     = 108.00
   ! n     = 1
! case  = Example 1

One way of  dealing with the input is to modify it by making it simpler. Below are the first two 
lines of  the input file above modified (and removing the descriptive text):

! 103165  10800 001
! 011265  10000

This essentially simplifies the FORMAT statement associated with the read. 

! 10  read (*,20) month, day, year, v, n
! 20  format (I2, I2, I2, X, F6.2, X, I3)

The second READ statement is modified in a similar manner:

!     read (*,20) month, day, year, p(j)

For the present, we will redirect input to the file, rather than dealing with file I/O.

! ./a.out < input.txt 

With the output, we will modify the last write statement. In its current form, 

      write (*, 320) i, r, case
      go to 10
  320 format (1h, 2f8.5, 9a6//)

It can be modified in the following manner:

      write (*, 320) i, r



      go to 10
  320 format (2f8.5, 2f8.5)

See listing rates_3.for

EXTERNAL FUNCTIONS

The next thing to note is that the program uses an external assembler function called TIME to 
convert the day, month, and year of  a specific date to decimal notation. This function can be re-
engineered in Fortran. We have done this by creating a separate file to contain the TIME 
function. This function can be written directly in Fortran 95, and stored in an external file 
(dtime.f95), and integrated using the interface specification. The function name has been 
changed to dtime, as time is built-in function. Specifications for this function can be found in 
Appendix A. The code within the program to access dtime is:

      interface
          function dtime(month,day,year)
              real dtime
              integer, intent(in) :: month, day, year
          end function dtime
      end interface

VARIABLES

The next portion of  the re-engineering involves modifying variables to remove implicit 
declarations, and enhance the declaration criteria to F95 standards. Variables that are implicitly 
declared in the original program are converted to explicit declarations. For example, arrays are 
converted int he following manner:

! dimension t difr(999), p(999), case(9)

becomes

! real, dimension(999) :: difr, p
! character, dimension(9) :: case

Variables previously declared in an intrinsic manner, and now also explicitly declared:

! integer :: i, j

See listing rates_4.for

LOOPS

The program contains three loops. The first of  these deals with reading a particular case from 
the file.



!     do 30 j = 1, n
!     read (*,20) month, day, year, p(j) 
! 30  difr(j)= t - time(month, day, year)

This is essentially invokes the READ command N times, storing the associated information in the 
array. This can be converted to F95 format in the following manner:

! do j = 1, n
!     read (*,20) month, day, year, p(j)
!     difr(j) = t - dtime(month, day, year)
! end do

The remaining two loops are nested, and deal with processing the information to derive an 
estimate of  the rate of  return.

      do 200 k = 1, limit
      val = v
      denom = 0.
      do 100 j = 1, n
      prsntv = p(j) * exp(i * difr(j))
      val = val - prsntv
  100 denom = denom + difr(j) * prsntv
      i = val/denom
      i = i + i
      if(abs(i) - 0.00005) 300, 300, 200
  200 continue

These loops are converted in a similar manner (with indenting added as modifications are made 
to improve readability):

      do k = 1, limit
          val = v
          denom = 0.
          do j = 1, n
              prsntv = p(j) * exp(i * difr(j))
              val = val - prsntv
              denom = denom + difr(j) * prsntv
          end do
          i = val/denom
          i = i + i
          if(abs(i) - 0.00005) 300, 300, 200
  200 end do

The label 200 is maintained until the arithmetic if is reconstructed.

See listing rates_5.for



ARITHMETIC IF

There is only one if statement, and unfortunately it is an arithmetic if. 

! if(abs(i) - 0.00005) 300, 300, 200

Basically the if statement branches three ways, depending on whether the result of  the expression 
is less than zero, zero, or greater than zero. For the interim, this can be converted in the following 
manner:

! if(abs(i) - 0.00005 <= 0) then
         go to 300
     end if

This can be improved upon by transferring the statements at label 300 into the body of  the if 
statement. These statements deal with calculating and outputting the value r.

         if (abs(i) - 0.00005 <= 0) then
              r = exp(i) - 1.0
              write (*, 320) i, r
  320         format (2f8.5, 2f8.5)
              exit
          end if

Doing this makes the go to statement following label 300 redundant. 

There is also an issue with the write statement which is invoked when the algorithm fails to 
converge. This appeared in the original code as:

      write output tape 6, 210, case
      go to 10  
 210  format (17H Failure on case 9A6)

It was invoked when the outer do loop ended naturally after all its iterations were expended.   
The use of  go to statements makes jumping around programs easy, but does lead to spaghetti-
like code.

We fix this by adding an integer variable success, which is initially 0 and set to when 1 when the 
algorithm converges. An if statement is then used after the “work” loop to prompt an error if  the 
algorithm did not converge for a particular case.

! success = 0
! limit = 100
! ...
! do k = 1, limit
         ....
!     ....
         if (abs(newi-i) - 0.00005 <= 0) then



!         r = exp(i) - 1.0
!         write (*, 320) i, r
 320         format (2f8.5, 2f8.5)
             success = 1
             exit
!     end if
! end do
! if (success == 0) then
!     write(*,*) "Failure on case"
! end if

See listing rates_6.for

FILE EXTENSIONS

At this point we can move the program from the .for extension, which signifies fixed format, 
to .f95, which signifies non-fixed formatting. This means we can indent some of  the labels 
associated with the format statements to improve readability.

PROGRAM TESTING

At this point, before we remove the go to statements, we should see if  the program is at a point 
where it is working. Compiling it now does not produce an exact result. Analyzing the original 
code further shows that this is likely the result of  the keywords which were removed earlier, 
namely BIG, SMALL, SUM, DELTA. Failing to find any information on how these keywords 
worked, it is time to go and look at Equation 17 from the original paper.

! ik+1 = ik +
VT �

P
Pjeik(T�tj)

P
(T � tj)Pjeik(T�tj)

where

	 Pj = the amount of  the jth payment into the fund
	 tj = the date of  the jth payment
	 r = the rate of  return, compounding annually.
	 V = the assets of  the fund (at market)
	 T = the date to which rate of  return is to be found
	 ik = the kth approximation of  the true rate of  return

This equation is applied until the difference between ik+1 and ik is smaller than some tolerance. 
In this algorithm the calculation is carried out to the point where |ik+1 - ik| < 0.00005. This 
calculation is performed using the statement:

! if (abs(i) - 0.00005 <= 0) then



which clearly does not calculate any difference value. Alluding to the original code:

! IF (ABSF(DELTA I) - 0.00005) 300, 300, 200

it is clear that DELTA I refers to some means of  deriving a differential value. Also the two 
statements prior to this (in the original code) do not seem quite logical:

! DELTA I = SUM VAL /DENOM
! SMALL I = SMALL I + DELTA I

Therefore the code should be re-written. The inner loop performs the summation, which is 
common to both the numerator and denominator. However in the existing code, there seems to 
be a discourse in the next portion of  code which calculates the new value of  i. To make the code 
work, we create a new real variable newi, which stores the value calculated by adding the current 
value of  i with the value calculated in the loop. This new value is then subtracted from the old 
and compared to the tolerance. 

Non-functioning code Functioning code

 do k = 1, limit
     val = v
     denom = 0.0
     do j = 1, n
         prsntv = p(j) * exp(i * difr(j))
         val = val - prsntv
         denom = denom + difr(j) * prsntv
     end do
     i = val/denom
     i = i + i

     if (abs(i) - 0.00005 <= 0) then
         r = exp(i) - 1.0
         write (*, 320) i, r
     320 format (2f8.5, 2f8.5)
         success = 1
         exit
      end if
 end do

 do k = 1, limit
     val = v
     denom = 0.0
     do j = 1, n
         prsntv = p(j) * exp(i * difr(j))
         val = val - prsntv
         denom = denom + difr(j) * prsntv
     end do
     newi = i + val/denom

     if (abs(newi-i) - 0.00005 <= 0) then
         r = exp(i) - 1.0
         write (*, 320) i, r
     320 format (2f8.5, 2f8.5)
         success = 1
         exit
     end if
     i = newi 
 end do

Using the values given in the paper, we test the program using the first series of  input:

! 103165  10800 001
! 011265  10000

The output produced by the program is:

! 0.09621 0.10099

which equates well with the sample output provided in the paper 0.09620, 0.10098.

See listing rates_7.f95



REMOVING GO TO

The purpose of  the go to statements is to transfer control back to the start of  the program to 
process the next set of  data. So if  the input is:

! 103165  10800 001
! 011265  10000
! 123165  13200 002
! 123164 -10000
! 123163  20000

the output becomes:

 ! 0.09621 0.10099
 ! 0.09531 0.10000

Now removing the final go to statement requires adding a third surrounding do loop. To process 
the input continuously, we modify the read statement, and add an if statement to exit the loop 
when there is no more input.

 ! do
 !    success = 0
         limit = 100
         read (*,20,iostat=ios) month, day, year, v, n
         20 format (i2, i2, i2, x, f6.2, x, i3)
         if (ios < 0) then
             exit
         end if
         ...
         ...
! end do

Here the read function sets the variable ios using the flag iostat. When the value of  ios 
becomes less than zero, there is no more input, and the program terminates.

See listing rates_8.f95

USABILITY

Now that the program clearly functions properly, there are some issues relating to usability which 
should be looked at. The first is the output, which is somewhat non-distinct. The values output 
represent the true rate of  return and the rate of  return compounding annually.
The output is currently in the form:

! 0.09621 0.10099
! 0.09531 0.10000
! 0.03966 0.04045



By modifying the write statement to include some context information, the output becomes 
more usable. Consider modifying the write and format statements in the following manner:

! write (*, 320) "True rate of return       = ", i*100, "%"
! write (*, 320) "Rate of return compounded = ", r*100, "%"
! 320 format (A,f8.5,A)
! write (*,*) " "

This produces output of  the form:

! True rate of return       =  9.62066%
! Rate of return compounded = 10.09866%
  
! True rate of return       =  9.53102%
! Rate of return compounded = 10.00000%
  
! True rate of return       =  3.96568%
! Rate of return compounded =  4.04537%

See listing rates_10.f95

VARIABLE NAMES

It may seem illogical to leave changing variable names to last, however leaving them as is makes it 
easier to re-engineering a program and refer back to the original code. Changing the variables 
only serves to improve their readability.

Original identifier New Identifier

v asset

p amntP

r rateCmpd

t rateTm

See listing rates_11.f95

http://archive.computerhistory.org/resources/text/Fortran/102665486.05.01.acc.pdf

http://archive.computerhistory.org/resources/text/Fortran/102665486.05.01.acc.pdf
http://archive.computerhistory.org/resources/text/Fortran/102665486.05.01.acc.pdf


FINAL NOTES

What does the final program look like?

!  modify file ending to f95 and change comment delimiter from C to !
!  sample application of rate of return algorithm (eq.17)
!
      program ratert

      interface
          function dtime(month,day,year)
              real dtime
              integer, intent(in) :: month, day, year
          end function dtime
      end interface

      real, dimension(999) :: difr, amntP
      real :: asset, rateTm, i, newi
      real :: val, denom, prsntv, rateCmpd 
      integer :: n, j, month, day, year
      integer :: limit, success, ios

      do 
          success = 0
          limit = 100
          read (*,20,iostat=ios) month, day, year, asset, n
      20  format (i2, i2, i2, x, f6.2, x, i3)
          if (ios < 0) then
              exit 
          end if
      
          rateTm = dtime(month, day, year)

          do j = 1, n
              read (*,20) month, day, year, amntP(j)
              difr(j) = rateTm - dtime(month, day, year)
          end do

          i = 0.0

          do k = 1, limit
              val = asset
              denom = 0.0
              do j = 1, n
                  prsntv = amntP(j) * exp(i * difr(j))
                  val = val - prsntv
                  denom = denom + difr(j) * prsntv
              end do
              newi = i + val/denom

              if (abs(newi-i) - 0.00005 <= 0) then
                  rateCmpd = exp(i) - 1.0
                  write (*, 320) "True rate of return       = ", i*100, "%"
                  write (*, 320) "Rate of return compounded = ", rateCmpd*100, "%"
                  320 format (A,f8.5,A)
                  write (*,*) " "
                  success = 1



                  exit
              end if
              i = newi
          end do
          if (success == 0) then
              write (*,*) "Failure on case"
          end if
      end do

      stop
      end 



APPENDIX A: The dtime function

The purpose of  the F95 function dtime is to convert from the month/day/year format to a 
decimal format. For example 3/25/2011 becomes 2011.2301. This makes it easier to calculate 
date differentials.

real function dtime(month, day, year)
    implicit none
    integer, intent(in) :: month, day, year
    integer :: leap_year, day_of_year, i

    real :: dg_date

    day_of_year = day

! Check for leap year, and add an extra day
    if (mod(year,400) == 0) then
        leap_year = 1
    else if (mod(year,100) == 0) then
        leap_year = 0
    else if (mod(year,4) == 0) then
        leap_year = 1
    else
        leap_year = 0
    end if

! Calculate the day of year
    do i = 1, month-1
        select case(i)
        case (1,3,5,7,8,10,12)
            day_of_year = day_of_year + 31
        case (4,6,9,11)
            day_of_year = day_of_year + 30
        case (2)
            day_of_year = day_of_year + 28 + leap_year
        end select
    end do

! Calculate the digital date in terms of fractional years
    if (leap_year == 0) then
        dg_date = year + (day_of_year / 365.0)
    else
        dg_date = year + (day_of_year / 366.0)
    end if

    dtime = dg_date

end function dtime 


